﻿<%
Dim strSourceFile,objXML,objRootList,rootOldNode,rootNewNode,XMLnode,objXML2,objConfig,iii,thisid
Dim tmp_title,tmp_link,tmp_time,tmp_info,tmp_image,operateact
Dim KinSlideshow
Dim brstr:brstr = Chr(13)&Chr(10)&Chr(9)

Function del()
	tmp_time = Request.QueryString("tmp_time")
	strSourceFile = Server.MapPath("bcastr.xml")
	Set objXML = Server.CreateObject("Microsoft.XMLDOM")
	objXML.load(strSourceFile)
	Set objRootList = objXML.documentElement.selectSingleNode("channel")
	For iii = 0 To objRootList.childNodes.Length -1
		If tmp_time = Trim(objRootList.childNodes.Item(iii).childNodes.Item(3).text) Then
			thisid = iii
			Exit For
		End If
	Next
	objRootList.removeChild(objRootList.childNodes(thisid))
	objXML.save(strSourceFile)
	Set objXML = Nothing

	Call Bcastr_SaveFunctions("Bcastr","Bcastr幻灯片","divBcastr","div",Bcastr_Html)
	Call SetBlogHint_Custom("√ 删除成功,侧栏模块已更新.")
	Response.Redirect "main.asp"
End Function

Function Save()
	tmp_title = Request.Form("tmp_title")
	tmp_link = Request.Form("tmp_link")
	tmp_image = Request.Form("tmp_image")
	tmp_time = Request.Form("tmp_time")

	operateact = Request.QueryString("operateact")

	strSourceFile = Server.MapPath("bcastr.xml")
	Set objXML = Server.CreateObject("Microsoft.XMLDOM")
	objXML.load(strSourceFile)
	Set objRootList = objXML.documentElement.selectSingleNode("channel")
	If operateact <> "update" Then 'INSERT
		XMLnode = brstr&"<item>"&brstr & _
				  "<title>"&tmp_title&"</title>"&brstr & _
				  "<link>"&tmp_link&"</link>"&brstr & _
				  "<image>"&tmp_image&"</image>"&brstr & _
				  "<time>"&tmp_time&"</time>"&brstr & _
				  "</item>"&Chr(13)
		Set objXML2 = Server.CreateObject("Microsoft.XMLDOM")
		objXML2.loadXML(XMLnode)
		Set rootOldNode = objRootList.childNodes.Item(0)
		Set rootNewNode = objXML2.documentElement
		'添加到节点最前
		objRootList.insertBefore rootNewNode, rootOldNode
	Else
		For iii = 0 To objRootList.childNodes.Length -1
			If tmp_time = Trim(objRootList.childNodes.Item(iii).childNodes.Item(3).text) Then
				thisid = iii
				Exit For
			End If
		Next
		Set rootOldNode = objRootList.childNodes.Item(thisid)
		XMLnode = brstr&"<item>"&brstr & _
				  "<title>"&tmp_title&"</title>"&brstr & _
				  "<link>"&tmp_link&"</link>"&brstr & _
				  "<image>"&tmp_image&"</image>"&brstr & _
				  "<time>"&tmp_time&"</time>"&brstr & _
				  "</item>"&Chr(13)
		Set objXML2 = Server.CreateObject("Microsoft.XMLDOM")
		objXML2.loadXML(XMLnode)
		Set rootNewNode = objXML2.documentElement
		objRootList.ReplaceChild rootNewNode, rootOldNode
	End If
	objXML.save(strSourceFile)
	Set objXML = Nothing
	Set objXML2 = Nothing

	Call Bcastr_SaveFunctions("Bcastr","Bcastr幻灯片","divBcastr","div",Bcastr_Html)
	Call SetBlogHint_Custom("√ 操作成功,侧栏模块已更新.")
	Response.Redirect "main.asp?act=list"
End Function

Function Edit()
	tmp_time = Request.QueryString("tmp_time")
	If Len(tmp_time) = 0 Then 'NEW
		operateact = "addnew"
		tmp_title = ""
		tmp_link = ""
		tmp_time = now()
		tmp_image= ""
	Else
		operateact = "update"
		strSourceFile = Server.MapPath("bcastr.xml")
		Set objXML = Server.CreateObject("Microsoft.XMLDOM")
		objXML.load(strSourceFile)
		Set objRootList = objXML.documentElement.selectSingleNode("channel")
		For iii = 0 To objRootList.childNodes.Length -1
			If tmp_time= Trim(objRootList.childNodes.Item(iii).childNodes.Item(3).text) Then
				tmp_title = objRootList.childNodes.Item(iii).childNodes.Item(0).text
				tmp_link = objRootList.childNodes.Item(iii).childNodes.Item(1).text
				tmp_image = objRootList.childNodes.Item(iii).childNodes.Item(2).text
				Exit For
			End If
		Next
		Set objXML = Nothing
	End If
	If operateact = "update" Then
	Response.Write "<div class=""divHeader"">幻灯片编辑</div>"
	Else
	Response.Write "<div class=""divHeader"">幻灯片添加</div>"
	End If
	Response.Write "<div class=""SubMenu""><a href=""?act=list""><span class=""m-left"">列表</span></a><a href=""?act=Edit""><span class=""m-left  m-now"">添加</span></a><a href=""?act=Config""><span class=""m-left"">设置</span></a></div>"
	Response.Write "<div id=""divMain2"">"
	Response.Write "<form id=""edit"" method=""post"">"
	Response.Write "<p>标题: <input id=""tmp_title"" name=""tmp_title"" style=""width:50%"" type=""text"" value="""& tmp_title &""" />(*) 　 <a onclick=""window.open(this.href,'','resizable=yes,width=810,height=600,status'); return false"" href=""insert.asp?from=article"">从文章插入</a></p>"
	Response.Write "<p>图片: <input id=""tmp_image"" name=""tmp_image"" style=""width:50%"" type=""text"" value="""& tmp_image &""" />(*) <a id=""helpbastr_"" href=""#"" title=""Bcastr不支持跨域图片，请使用站内图片"">？</a> <a onclick=""window.open(this.href,'','resizable=yes,width=810,height=600,status'); return false"" href=""insert.asp?from=upload"">从附件插入</a>"
	If CheckPluginState("windsphoto")=True Then Response.Write " <a onclick=""window.open(this.href,'','resizable=yes,width=810,height=600,status'); return false"" href=""insert.asp?from=windsphoto"">从WindsPhoto插入</a>"
	Response.Write "</p>"
	Response.Write "<p>链接: <input id=""tmp_link"" name=""tmp_link"" style=""width:50%"" type=""text"" placeholder=""http://"" value="""& tmp_link &""" />(*)</p>"
	Response.Write "<input id=""tmp_time"" name=""tmp_time""  type=""hidden"" value="""& tmp_time &""" />"
	Response.Write "<p><input type=""submit"" class=""button"" value="""& ZC_MSG087 &""" id=""btnPost"" onclick='return checkpost();' /></p>"

	Response.Write "</form>"
	Response.Write "</div>"
%>
<script>
function checkpost(){
	document.getElementById("edit").action="main.asp?act=Save&operateact=<%=operateact%>";
	if(!document.getElementById("tmp_title").value){
		alert('请填写图片标题!');
		return false
	}
	if(!document.getElementById("tmp_image").value){
		alert('请填写图片地址!');
		return false
	}
	if(!document.getElementById("tmp_link").value){
		alert('请填写图片链接!');
		return false
	}
}
</script>
<%
End Function

Function List()
	Response.Write "<div class=""divHeader"">幻灯片列表</div>"
	Response.Write "<div class=""SubMenu""><a href=""?act=list""><span class=""m-left m-now"">列表</a><a href=""?act=Edit""><span class=""m-left"">添加</span></a><a href=""?act=Config""><span class=""m-left"">设置</span></a></div>"
	Response.Write "<div id=""divMain2"">"
	strSourceFile = Server.MapPath("bcastr.xml")
	Set objXML = Server.CreateObject("Microsoft.XMLDOM")
	objXML.load(strSourceFile)
	Set objRootList = objXML.documentElement.selectSingleNode("channel")
	For iii = 0 To objRootList.childNodes.Length -1
		tmp_title = objRootList.childNodes.Item(iii).childNodes.Item(0).text
		tmp_link = objRootList.childNodes.Item(iii).childNodes.Item(1).text
		tmp_image = objRootList.childNodes.Item(iii).childNodes.Item(2).text
		tmp_time = objRootList.childNodes.Item(iii).childNodes.Item(3).text

		Response.Write "<div style=""width:25%;text-align:center;float:left""><a href="&tmp_link&" target=_blank title="&tmp_title&"><img src="&tmp_image&" alt="&tmp_title&" style=""max-width:200px;max-height:200px;padding-bottom:3px;""></a><br><a href=""?act=Edit&tmp_time=" & tmp_time & """>[编辑]</a> <a href=""?act=del&tmp_time=" & tmp_time & """ onclick=""return window.confirm('你将在幻灯片中删除此图片,确定吗?');"">[删除]</a></div>"

	Next
	Set objXML = Nothing

	Response.Write "<div style=""clear:both;margin:50px 0 0 50px;""><p>预览：</p><p>"&Bcastr_Html&"</p></div>"
	Response.Write "</div>"
End Function

Function SaveConfig()
	Dim strBCASTR_WIDTH,strBCASTR_HEIGHT,strBCASTR_BCASTR,strBCASTR_KINSLIDESHOW
	strBCASTR_WIDTH = Request.Form("strBCASTR_WIDTH")
	strBCASTR_HEIGHT = Request.Form("strBCASTR_HEIGHT")
	strBCASTR_BCASTR = Request.Form("strBCASTR_BCASTR")
	strBCASTR_KINSLIDESHOW = Request.Form("strBCASTR_KINSLIDESHOW")
	BCASTR_BCASTR = CBool(strBCASTR_BCASTR)
	BCASTR_KINSLIDESHOW = CBool(strBCASTR_KINSLIDESHOW)
	
	strSourceFile = Server.MapPath("bcastr.xml")
	Set objXML = Server.CreateObject("Microsoft.XMLDOM")
	objXML.load(strSourceFile)
	Set objRootList = objXML.documentElement
	Set rootOldNode = objRootList.childNodes.Item(0)
	str=Array("")
	For iii=0 to rootOldNode.childNodes.Length -1
		Redim Preserve str(iii)
		str(iii)=Replace(Request.form("str"&iii),"#","0x")
	Next
	
	'检查提交表单
	If IsNumeric(strBCASTR_HEIGHT) = False Or IsNumeric(str(0)) = False Or IsNumeric(str(1)) = False Or IsNumeric(str(4)) = False Or IsNumeric(str(7)) = False Or IsNumeric(str(10)) = False Or IsNumeric(str(12)) = False Or IsNumeric(str(13)) = False Or Instr(str(8),"0x") = "0" Or Instr(str(9),"0x") = "0" Or Instr(str(14),"0x") = "0" Or Instr(str(15),"0x") = "0" Or Instr(str(16),"0x") = "0" Or Instr(str(16),"0x") = "0" Or Instr(str(17),"0x") = "0" Then ShowError(3)
	
	XMLnode = brstr&"<config>"&brstr & _
		"<roundCorner>"&str(0)&"</roundCorner>"&brstr & _
		"<autoPlayTime>"&str(1)&"</autoPlayTime>"&brstr & _
		"<isHeightQuality>"&LCase(str(2))&"</isHeightQuality>"&brstr & _
		"<blendMode>"&str(3)&"</blendMode>"&brstr & _
		"<transDuration>"&str(4)&"</transDuration>"&brstr & _
		"<windowOpen>"&str(5)&"</windowOpen>"&brstr & _
		"<btnSetMargin>"&str(6)&"</btnSetMargin>"&brstr & _
		"<btnDistance>"&str(7)&"</btnDistance>"&brstr & _
		"<titleBgColor>"&str(8)&"</titleBgColor>"&brstr & _
		"<titleTextColor>"&str(9)&"</titleTextColor>"&brstr & _
		"<titleBgAlpha>"&str(10)&"</titleBgAlpha>"&brstr & _
		"<titleFont>"&str(11)&"</titleFont>"&brstr & _
		"<titleMoveDuration>"&str(12)&"</titleMoveDuration>"&brstr & _
		"<btnAlpha>"&str(13)&"</btnAlpha>"&brstr & _
		"<btnTextColor>"&str(14)&"</btnTextColor>"&brstr & _
		"<btnDefaultColor>"&str(15)&"</btnDefaultColor>"&brstr & _
		"<btnHoverColor>"&str(16)&"</btnHoverColor>"&brstr & _
		"<btnFocusColor>"&str(17)&"</btnFocusColor>"&brstr & _
		"<changImageMode>"&str(18)&"</changImageMode>"&brstr & _
		"<isShowBtn>"&LCase(str(19))&"</isShowBtn>"&brstr & _
		"<isShowTitle>"&LCase(str(20))&"</isShowTitle>"&brstr & _
		"<scaleMode>"&str(21)&"</scaleMode>"&brstr & _
		"<transform>"&str(22)&"</transform>"&brstr & _
		"<isShowAbout>false</isShowAbout>"&brstr & _
		"</config>"&Chr(13)
	Set objXML2 = Server.CreateObject("Microsoft.XMLDOM")
	objXML2.loadXML(XMLnode)
	Set rootNewNode = objXML2.documentElement
	'保存xml设置
	objXML.documentElement.ReplaceChild rootNewNode, rootOldNode
	objXML.save(strSourceFile)
	Set objXML = Nothing
	Set objXML2 = Nothing
	'保存插件基本设置
	Bcastr_Config.Write "BCASTR_WIDTH",strBCASTR_WIDTH
	Bcastr_Config.Write "BCASTR_HEIGHT",strBCASTR_HEIGHT
	Bcastr_Config.Write "BCASTR_BCASTR",strBCASTR_BCASTR
	Bcastr_Config.Write "BCASTR_KINSLIDESHOW",strBCASTR_KINSLIDESHOW
	Call Bcastr_Config.Save
	
	Dim intervalTime,moveSpeedTime,moveStyle,mouseEvent,isHasTitleBar,isHasTitleFont,isHasBtn,TitleFont_color,TitleFont_family,titleBar_bgColor,titleBar_alpha,btn_bgColor,btn_bgHoverColor,btn_fontColor,btn_bgAlpha,strKinSlideShowConfig
	intervalTime = str(1)
	moveSpeedTime = str(4)*1000
	moveStyle = Replace(str(22),"bottom","down")
	mouseEvent = Replace(Replace(str(18),"click","mouseclick"),"hover","mouseover")
	isHasTitleBar = LCase(str(20))
	isHasTitleFont = LCase(str(20))
	isHasBtn = LCase(str(19))
	TitleFont_color = Replace(str(9),"0x","#")
	TitleFont_family = str(11)
	titleBar_bgColor = Replace(str(8),"0x","#")
	titleBar_alpha = str(10)
	btn_bgColor = Replace(str(15),"0x","#")
	btn_bgHoverColor = Replace(str(16),"0x","#")
	btn_fontColor = Replace(str(14),"0x","#")
	btn_bgAlpha = str(13)
	
	strKinSlideShowConfig = "$(function(){$(""#KinSlideshow"").KinSlideshow({intervalTime:"&intervalTime&",moveSpeedTime:"&moveSpeedTime&",moveStyle:"""&moveStyle&""",mouseEvent:"""&mouseEvent&""",isHasBtn:"&isHasBtn&",isHasTitleBar:"&isHasTitleBar&",isHasTitleFont:"&isHasTitleFont&",titleFont:{TitleFont_size:12,TitleFont_color:"""&TitleFont_color&""",TitleFont_family:"""&TitleFont_family&""",TitleFont_weight:""normal""},titleBar:{titleBar_height:20,titleBar_bgColor:"""&titleBar_bgColor&""",titleBar_alpha:"&titleBar_alpha&"},btn:{btn_bgColor:"""&btn_bgColor&""",btn_bgHoverColor:"""&btn_bgHoverColor&""",btn_fontColor:"""&btn_fontColor&""",btn_fontHoverColor:"""&btn_bgColor&""",btn_fontFamily:"""&TitleFont_family&""",btn_borderColor:"""&btn_bgColor&""",btn_borderHoverColor:"""&btn_bgHoverColor&""",btn_borderWidth:1,btn_bgAlpha:"&btn_bgAlpha&"}})})"
	'如果启用KinSlideShow，保存KinSlideShow设置javascript
	If BCASTR_KINSLIDESHOW = True Then Call SaveToFile(BlogPath & "zb_users/plugin/bcastr/KinSlideshow.config.js", strKinSlideShowConfig, "utf-8", False)
	'重新生成模块
	Call Bcastr_SaveFunctions("Bcastr","Bcastr幻灯片","divBcastr","div",Bcastr_Html)
	Call SetBlogHint(True,False,False)
	Response.Redirect "main.asp?act=Config"
End Function

Function Config()
	strSourceFile = Server.MapPath("bcastr.xml")
	Set objXML = Server.CreateObject("Microsoft.XMLDOM")
	objXML.load(strSourceFile)
	Set objRootList = objXML.documentElement.selectSingleNode("config")
	Dim tmp_item,tmp_value,tmp_type,tmp_extra
	'定义设置项目数组
	tmp_item=Array("<b> · 图片圆角</b> <br /><span class='note'>仅对Bcastr有效</span>","<b> · 图片切换间隔时间</b> <br /><span class='note'>单位秒</span>","<b> · 图片高质量缩小</b> <br /><span class='note'>仅对Bcastr有效</span>","<b> · 图片混合模式</b> <br /><span class='note'>仅对Bcastr有效</span>","<b> · 切换一张图片所需的时间</b> <br /><span class='note'>单位秒</span>","<b> · 图片链接打开方式</b> <br /><span class='note'>”_blank”在新窗口打开，也可以使用”_self”在本窗口打开，仅对Bcastr有效，KinSlideShow默认新窗口打开</span>","<b> · 按钮的位置</b> <br /><span class='note'>按钮文字的位置，设置方法同css的margin，仅对Bcastr有效</span>","<b> · 每个按钮的距离</b> <br /><span class='note'>仅对Bcastr有效</span>","<b> · 标题背景的颜色</b>","<b> · 标题文字的颜色</b>","<b> · 标题背景的透明度</b>","<b> · 标题文字的字体</b>","<b> · 标题背景动画的时间</b> <br /><span class='note'>仅对Bcastr有效，单位秒</span>","<b> · 按钮的透明度</b>","<b> · 按钮文字的颜色</b>","<b> · 按钮颜色</b>","<b> · 按钮的悬停颜色</b> <br /><span class='note'>仅对Bcastr有效</span>","<b> · 按钮当前颜色</b>","<b> · 切换图片的方法</b> <br /><span class='note'>”click”点击切换，”hover”鼠标悬停切换</span>","<b> · 是否显示按钮</b>","<b> · 是否显示标题</b>","<b> · 图片缩放模式 </b> <br /><span class='note'>“showAll”:可以看到全部图片,保持比例,可能上下或者左右，“exactFil”:放缩图片到舞台的尺寸,可能比例失调，“noScale”:图片的原始尺寸,无放缩，“noBorder”:图片充满播放器,保持比例,可能会被裁剪<br />仅对Bcastr有效，<b>注意大小写</b>，KinSlideShow默认为exactFil</span>","<b> · 图片转换模式</b> <br /><span class='note'>“alpha”:透明度淡入淡出，“blur”:模糊淡入淡出，“left”:向左切换，“right”:向右切换，“top”:向上切换，“bottom”:向下切换，“breathe”:有一点点地放缩的淡入淡出，“breatheBlur”:有一点点地放缩的模糊淡入淡出<br />仅上下左右对KinSlideShow有效，默认值为left</span>","<b> · 显示关于信息</b>")
	Response.Write "<div class=""divHeader"">幻灯片设置</div>"
	Response.Write "<div class=""SubMenu""><a href=""main.asp""><span class=""m-left"">列表</span></a><a href=""?act=Edit""><span class=""m-left"">添加</span></a><a href=""?act=Config""><span class=""m-left m-now"">设置</span></a></div>"
	Response.Write "<div id=""divMain2"">"
	Response.Write "<form name=""edit"" method=""post"" action=""main.asp?act=SaveConfig"">"
	Response.Write "<table width=""100%"" style=""padding:0px;margin:1px;"" cellspacing=""0"" cellpadding=""0"">"
	Response.Write "<tr><td style=""width:32%""><p align=""left""><b> · 幻灯片大小设置</b> <br /><span class='note'>如果在侧栏调用，宽度请根据侧栏进行调整。宽度可填写数字或百分比，高度请填写数字固定</span></p></td><td><input name=""strBCASTR_WIDTH"" style=""width:70px"" type=""text"" value="""&BCASTR_WIDTH&""" />x<input name=""strBCASTR_HEIGHT"" style=""width:70px"" type=""text"" value="""&BCASTR_HEIGHT&""" /></td></tr>"
	Response.Write "<tr><td><p align=""left""><b> · Bcastr开关</b></p></td><td><input type=""text"" name=""strBCASTR_BCASTR"" class=""checkbox"" value="""&BCASTR_BCASTR&""" /></td></tr>"
	Response.Write "<tr><td><p align=""left""><b> · KinSlideShow开关</b></p></td><td><input type=""text"" name=""strBCASTR_KINSLIDESHOW"" class=""checkbox"" value="""&BCASTR_KINSLIDESHOW&""" /></td></tr>"
	For iii = 0 To objRootList.childNodes.Length -2
		tmp_value = objRootList.childNodes.Item(iii).text
		'处理xml设置值及input type等
		If Instr(tmp_value,"0x") Then
			tmp_type = "color"
			tmp_value = Replace(tmp_value,"0x","#")
			tmp_extra = ""
		ElseIf Instr(tmp_value,"true") Or Instr(tmp_value,"false") Then
			tmp_type = "text"
			tmp_value = Replace(tmp_value,"true","True")
			tmp_value = Replace(tmp_value,"false","False")
			tmp_extra = "class='checkbox'"
		Else
			tmp_type = "text"
			tmp_extra = "style='width:90%'"
		End If
		Response.Write "<tr><td><p align=""left"">"&tmp_item(iii)&"</p></td><td><input type="""&tmp_type&""" "&tmp_extra&" name=""str"&iii&""" value="""&tmp_value&""" /></td></tr>"
	Next
	Response.Write "</table><p><input type=""submit"" class=""button"" value=""提交"" id=""btnPost"" /><input type=""reset"" class=""button"" value=""重置"" id=""btnPost"" /></p></form>"
	Set objXML = Nothing
End Function

Function Bcastr_Html
	strSourceFile = Server.MapPath("bcastr.xml")
	Set objXML = Server.CreateObject("Microsoft.XMLDOM")
	objXML.load(strSourceFile)
	Set objRootList = objXML.documentElement.selectSingleNode("channel")
	For iii = 0 To objRootList.childNodes.Length -1
		tmp_title = objRootList.childNodes.Item(iii).childNodes.Item(0).text
		tmp_link = objRootList.childNodes.Item(iii).childNodes.Item(1).text
		tmp_image = objRootList.childNodes.Item(iii).childNodes.Item(2).text
		tmp_time = objRootList.childNodes.Item(iii).childNodes.Item(3).text
		KinSlideshow = KinSlideshow & "<a href="""&tmp_link&""" target=""_blank"" title="""&tmp_title&"""><img src="""&tmp_image&""" width="""&BCASTR_WIDTH&""" height="""&BCASTR_HEIGHT&""" alt="""&tmp_title&""" /></a>"
	Next
	Set objXML = Nothing
	If BCASTR_KINSLIDESHOW = True Then
		Bcastr_Html = Bcastr_Html & "<script src="""&ZC_BLOG_HOST&"zb_users/plugin/bcastr/jquery.KinSlideshow-1.2.1.min.js"" type=""text/javascript""></script>" & vbCrLf
		Bcastr_Html = Bcastr_Html & "<script src="""&ZC_BLOG_HOST&"zb_users/plugin/bcastr/KinSlideshow.config.js"" type=""text/javascript""></script>" & vbCrLf
		Bcastr_Html = Bcastr_Html & "<style>#divBcastr div{padding:0;margin:0;}#divBcastr ul{border-top:0;width:auto;}#divBcastr li{padding:0;}#KinSlideshow .KSS_titleBar, #KinSlideshow .KSS_titleBox{bottom:auto;top:0 !important;} #KinSlideshow .title{text-align:center;}</style>" & vbCrLf
	End If
	If BCASTR_BCASTR = True Then
		Bcastr_Html = Bcastr_Html & "<object id=""bcastr4"" data="""&ZC_BLOG_HOST&"zb_users/plugin/bcastr/bcastr4.swf?xml="&ZC_BLOG_HOST&"zb_users/plugin/bcastr/bcastr.xml"" type=""application/x-shockwave-flash"" width="""&BCASTR_WIDTH&""" height="""&BCASTR_HEIGHT&""">" & vbCrLf
		Bcastr_Html = Bcastr_Html & "<param name=""movie"" value="""&ZC_BLOG_HOST&"zb_users/plugin/bcastr/bcastr4.swf?xml="&ZC_BLOG_HOST&"zb_users/plugin/bcastr/bcastr.xml"" />" & vbCrLf
	End If
	If BCASTR_KINSLIDESHOW = True Then
		Bcastr_Html = Bcastr_Html & "<div id=""KinSlideshow"" style=""visibility:hidden;"">"&KinSlideshow&"</div>"
	End If
	If BCASTR_BCASTR = True Then
		Bcastr_Html = Bcastr_Html & "</object>"
	End If
End Function
%>